# ECE 375 Lab 8

Morse Code Transmitter

Lab Time: Wednesday 10a-12n

Robert Detjens

David Headrick

#### Introduction

In this lab, we used two main routines, endlessly looping in our MAIN routine to carry out the requested functionality. We used polling to listen for button inputs, and Timer/Counter 1 to keep track of time during the morse code segments. We used a jump table to keep record of every letter of the alphabet, and its associated morse code sequence. One major difficulty we had was the jump table not working because it kept outputting the letter A, but other than that, this lab was mostly smooth sailing.

## **Program Overview**

Three main routines were used: INTRO, PROMPT, and MORSE.

INTRO displayed a welcome text, and asks the user to press a Button 0 to start, then blocks until that button is pressed.

PROMPT is the main runner function for getting the user input string. It records button presses and store the corresponding ASCII text into data memory and displays it on the LCD. Polling was used to check for button presses. To keep track of the current letter, a specific register was used, then when Button 0 is pressed, this register is stored to the next character in LCDLine2 (tracked with Y and a post-increment), then resets back to the default letter of A for the next letter selection. Once the line is full or when Button 4 is pressed, the line is space-terminated and this procedure returns.

MORSE is used to translate the string of characters filled from PROMPT in data memory into Morse code output on LEDs 7:5. This procedure uses a jump table and and indirect jump to efficiently perform the correct Morse sequence for each letter, instead a sequence of comparisons. The start address of this jump table is defined using a label. Each block in the jump table was padded to 5 instructions with nops, as the maximum length of a letter's morse code sequence is 4 symbols (4 realls to DOT/DASH), plus an additional word for a ret instruction. To calculate the offset into the jump table, the following formula was used:

$$JUMP\_TABLE + [(ascii\_letter - 'A') * 5]$$

As each block in the jump table is 5 words long, this formula will index into the jump table based on the "index" of the ascii number (0-26), multiplied by 5 for the 5-word padding. This resultant address is stored into Z for use with icall.

### **Difficulties**

Our main difficulties were in trying to get the jump table to work properly.

Initially, we used call instructions in the jump table to call the DOT and DASH functions. This was a problem because we were expecting each instruction in the table to be 16 bits in order to adhere to the 5-word padding. However, we forgot that the call instruction is a

two-word instruction and consumes 32 bits. Switching to using rcalls instead, which are 16 bits, fixed this alignment issue.

Before we jumped to the jump table, we calculated the offset into the jump table. We knew this would fit into 8 bits, so we just used a register to store the result. However, the final value of the address into the jump table – after adding the offset to the base address – was 16 bits. This ended up causing a 8 bit overflow when adding the offset to the jump table address, which we didn't account for. We ended up using ADD to add the offset and the low byte of base address, then propagating any overflow to the high byte of the address with ADC.

Initially, when using icall to perform the call into the jump table, we misinterpreted the instruction set manual. We thought that the Z register held a pointer to where the destination address was. This extra level of indirection was incorrect, and Z holds the destination address directly. Once we realized this, we loaded the destination directly into Z and the call worked correctly.

The worst difficulty we had was an off-by-one error in our jump table. When performing the morse code transmission, the first segment of the first letter of the message was always skipped. For example: AA would always print out DASH DOT DASH, missing the initial DOT. We never found the cause of this error, but we did bodge in a workaround. Before the message is printed, the letter E is printed first. E only has one segment (DOT), which is always skipped because of this bug. This ensures that the entire string the user submitted is printed.

In addition, while trying to solve the above off-by-one error, we added a NOP into the jump table to try and fix the off-by-one issue. This didn't fix it, and we forgot to remove it which caused more confusion down the line. We eventually remembered to remove it, and it worked after.

The simulator proved to be annoying during this lab. Anything that used a timer, e.g. the Morse code delays, took a very long time to simulate. We had to comment out a few lines and make other modifications to make debugging in the simulator easier. That being said, the examination tools in the simulator proved to be pivotal when debugging the mentioned issues with the jump table.

## Conclusion

The hardest part of Lab 8 was the jump table. The jump table took the longest amount of time to debug because of the odd errors.

As far as the button presses go, our past labs helped us out a lot. We were able to reuse polling code from previous labs. Not once in this lab did we have a problem with the buttons, just took some fine tuning for the debounce length.

Throughout the source code, we modularized functionality to keep procedures short, clean, and readable. We also used plenty of comments to document what everything was doing. As the lab was completed over the span of 2 weeks, these comments were useful to remember what the previously-written code was doing. Once we came back over thanksgiving break, the well documented, clean code helped us get right back to work.

All of our past labs helped us a lot. Overall, we used a lot of the same techniques used in previous labs. Button polling, Timer/Counter, and 16-bit add just to name a few. During the course of this whole lab class, we've been constantly referring to previous labs to remember how something is properly done.

#### Source Code

```
Robert Detjens & David Headrick Lab 8 Source Code
;*********************
;*
  Author: Robert Detjens
      David Headrick
;*
   Date: 11/30/21
:*
;*
.include "m128def.inc"
             ; Include definition file
; ==== REGISTER CONSTANTS
.def
           = r16
    mpr
.def
    curr letter
           = r23
.def
    zeroreg
           = r24
;*********************
;* Start of Code Segment
; beginning of code segment
; **********************************
;* Interrupt Vectors
.org $0000
rjmp INIT
        ; reset interrupt
         ; end of interrupt vectors
.org
   $0046
;* Program Initialization
```

```
INIT:
 ; Initialize the Stack Pointer
 ldi mpr, low(RAMEND)
 out SPL, mpr; Load SPL with low byte of RAMEND
 ldi mpr, high(RAMEND)
 out SPH, mpr; Load SPH with high byte of RAMEND
  ; Configure I/O ports
 ; Initialize Port B for output
 ldi mpr, $FF; Set Port B Directional Register
 out DDRB, mpr; for output
 ldi mpr, $00; Initialize Port B for outputs
 out PORTB, mpr; Port B outputs low
 ; Initialize Port D for input
 ldi mpr, $00; Set Port D Directional Register
 out DDRD, mpr; for inputs
 ldi mpr, $FF; Initialize Port D for inputs
 out PORTD, mpr; with pull-up
  ; init LCD
 call LCDInit
  ; configure Timer 1 for 1|3s sleep
  ; set timer to normal mode
  ; ldi
         mpr,
         TCCR1A,
  ; out
                   mpr
                 Ob00000101 ; clk/1024 prescaling
 ldi
       mpr,
  ; ldi
                   Ob00000100 ; TODO: remove, clk/256
         mpr,
       TCCR1B,
 out
                mpr
  ; ldi
         mpr,
  ; out
         TCCR1C,
                  mpr
  ; Enable global interrupts (if any are used)
  ; sei
  ; clear zero register
 clr
         zeroreg
  ; Display intro message & wait for button
 rcall
         INTRO
;* Main Program
```

```
MAIN:
 rcall
        PROMPT
 rcall
        MORSE
 rjmp
        MAIN
               ; return to top of MAIN
:* Functions
; INTRO()
   Displays a welcome message on the LCD
   and waits for the user to press button 0.
INTRO:
 ; load intro strings to lcd memory
 ; Move strings from Program Memory to Data Memory
 ; location of string in program memory
             low(WELCOME L1 S << 1)</pre>
 ldi
        ZL,
 ldi
        ZH,
             high(WELCOME L1 S << 1)
 ; dest addr in data memory (0x0100)
             low(LCD Line1)
 ldi
        YL,
 ldi
        YH,
             high(LCD_Line1)
 istr1 1:
   lpm
          mpr,
               Z+
          Y+,
   st
               mpr
   cpi
          YL, low(WELCOME_L1_E << 1)
          istr1 l
   brne
 ; String 2
             low(WELCOME_L2_S << 1)</pre>
 ldi
        ZL,
 ldi
             high(WELCOME L2 S << 1)
 ; dest addr in data memory (0x0100)
        YL,
             low(LCD Line2)
 ldi
 ldi
        YH,
             high(LCD_Line2)
 istr2 1:
          mpr,
               Z+
   lpm
   st
          Υ+,
               mpr
          YL, low(WELCOME_L2_E << 1)
   cpi
   brne
          istr2 1
 ; display the strings
```

```
call LCDWrite
  ; now wait for pd0 button
  wait_for_b0:
    ; get button inputs
                  PIND
          mpr,
    in
    sbrc mpr,
                  0
          wait_for_b0
    jmp
  rcall DEBOUNCE
  ret
; PROMPT()
    Main function for getting the word input from the user.
   Polls buttons:
             DASH 6/7 step through letters
      rcall DASH O confirms the letter
      rcall DASH 4 confirms the word and returns (for transmission)
PROMPT:
  ; load prompt string into line 1
  ; location of string in program memory
                low(PROMPT_S << 1)</pre>
  ldi
          ZL,
  ldi
          ZH,
                high(PROMPT S << 1)
  ; dest addr in data memory (0x0100)
  ldi
          YL,
                low(LCD Line1)
  ldi
          YH,
                high(LCD_Line1)
  str1_l:
    lpm
            mpr, Z+
    st
            Υ+,
                  mpr
            YL, low(PROMPT_E << 1)
    cpi
    brne
            str1 l
  ; display string
  call
          LCDWrLn1
  call
          LCDClrLn2
  ; set Y to line 2
  ; second line in data memory (0x0110)
          YL,
                low(LCD Line2)
  ldi
  ldi
          YH,
                high(LCD_Line2)
  ; turn on cursor on LCD
```

```
; no proc for this, so send command manually
                 0b00001110; 0 0 0 0 1 D C B
; ldi
         mpr,
; call
         {\tt LCDWriteCmd}
; start with A
       curr letter, 'A'
ldi
PROMPT LOOP:
  ; store current letter to LCD memory
         Y, curr_letter
  ; update LCD
  call LCDWrLn2
  ; get button inputs
       mpr,
               PIND
  in
  sbrs mpr,
              7; bit 7: decrement letter
       BIT 7
  jmp
              6; bit 6: increment letter
  sbrs mpr,
  jmp
       BIT 6
               0; bit 0: confirm letter
  sbrs mpr,
  jmp
       BIT_0
               4; bit 4: confirm whole word
  sbrs mpr,
       BIT 4
  jmp
       BIT_NONE ; skip bit handling
  jmp
 BIT 7:
   ; decrement letter
   dec curr_letter
    ; wrap around if it underflowed
   cpi curr letter, 64; 'A' is 65
   brne DEC NOOP
      ; only decrement if above min
           curr letter, 90 ; 'Z' is 90
     ldi
   DEC NOOP:
    jmp BIT_DONE
 BIT 6:
    ; increment letter
   inc curr_letter
    ; wrap around if it overflowed
   cpi curr letter,
                       91 ; 'z' is 90
   brne INC NOOP
     ; only decrement if above min
```

```
curr letter, 65; 'A' is 65
       ldi
      INC NOOP:
      jmp BIT_DONE
   BIT 0:
      ; letter confirmed, increment dest addr
      ; store a final time with post-increment
             Y+, curr_letter
      st
      ; if 16 chars have been entered, start transmission (button 4)
             YL,
                    low(LCD End)
     breq
             BIT 4
      ; start with A once more
      ldi
             curr letter, 65
      jmp BIT_DONE
   BIT 4:
      ; word confirmed, exit prompt
      ; but make sure the string is space-terminated first
           Υ+,
                         curr letter
           curr_letter, ''
      ldi
                         curr_letter
      st
           Υ+,
     ret
   BIT_DONE:
    rcall DEBOUNCE
   BIT NONE:
  ; keep looping until button 4 is hit
  jmp PROMPT LOOP
DEBOUNCE:
  ; wait for a small delay to do some real basic debouncing
  ; wait a bit for debounce
  ; reuse exising wait func for inner loop
 ldi
       mpr,
               100
 debounce 1:
   ldi wait, 255
   call LCDWait
```

```
dec
          mpr
  brne
       debounce l
  ret
; MORSE()
    Broadcasts the characters in data memory $1010:1020
    as Morse code over the top 3\ \text{LEDS} on port B
MORSE:
  ; turn on PIN/LED 4 to signal broadcasting
  ldi
        mpr,
                0b00010000
  out
        PORTB,
                mpr
  ; go through chars in line 2
  ; second line in data memory (0x0110)
  ldi
                low(LCD Line2)
          YL,
  ldi
          YH,
                high(LCD_Line2)
  ; for some reason the first dot/dash is ignored,
  ; so "display" a letter with only one dot (e.g. E)
  ; and now subsequent letters (the actual message) works
  ; I Love Programming:tm:
          curr letter,
  ldi
  morse loop:
    ; print current char
    rcall PRINT_MORSE
    ; load next one
    ld
          curr letter, Y+
    ; if the the next character is not ' ', loop back
    cpi
          curr letter,
    brne morse_loop
  ; turn off PIN/LED 4 when broadcast is done
  ldi
        mpr,
        PORTB, mpr
  out
  ret
; PRINT_MORSE():
    Prints the ascii char in curr_letter as morse code. This uses an jump into
    JUMP_TABLE to efficiently perform the correct sequence of dots/dashes based
    on an index calculated from the current letter.
PRINT_MORSE:
```

```
; curr_letter = (curr_letter - 'A') * 5
  subi curr_letter,
                     ' A '
  mov
       mpr, curr_letter
       curr letter
  lsl
       curr letter
  lsl
       curr_letter, mpr
  add
  ; load address of jump table into Z for indirect call
              LOW(JUMP_TABLE)
  ldi
  ldi
        ZH,
              HIGH(JUMP_TABLE)
  ; add calculated letter offset
  add
        ZL,
            curr letter
  adc
        ZH,
            zeroreg
  ; indirect call to JUMP TABLE+offset
  icall
  ; wait 2 more units (for 3 total) between letters
  rcall WAIT 1
  rcall WAIT_1
  ret
DOT:
  ; wait 1 unit on, 1 unit off
  ; turn on signal leds
                0b11110000
  ldi
        mpr,
        PORTB, mpr
  out
  rcall
          WAIT 1
  ; turn off signal pins
  ldi
                0b00010000
        mpr,
        PORTB, mpr
  out
  rcall
          WAIT_1
  ret
DASH:
  ; wait 3 units on, 1 unit off
  ; turn on signal leds
```

```
0b11110000
  ldi
       mpr,
       PORTB, mpr
  out
  rcall
          WAIT_3
  ; turn off signal pins
  ldi
        mpr,
                0b00010000
  out
        PORTB, mpr
         WAIT_1
  rcall
  ret
WAIT_1:
  ; 0xFFFF - 16000 (0x3E80) = 0xC17F
 ldi
                  0xC1
        mpr,
  out
        TCNT1H,
                  mpr
  ldi
                  08x0
        mpr,
  out
       TCNT1L,
                  mpr
  ; wait for timer overflow (reuse loop)
  rjmp wait_for_timer
WAIT 3:
  ; 0xFFFF - 48000 (0xBB800) = 0x447F
  ldi
                 0x44
        mpr,
  out
       TCNT1H,
                  mpr
  ldi
        mpr,
                  0x7f
  out
        TCNT1L,
                  mpr
  wait for timer:
    ; check TOV1 bit in TIFR flag register \,
                    TIFR
          mpr,
                    0b00000100
    andi mpr,
    breq wait_for_timer ; loop if not set
  ; clear overflow flag
                  0b00000100
  ldi
        mpr,
        TIFR,
  out
                  mpr
  ret
JUMP TABLE:
  rcall DOT
```

```
rcall
        DASH
ret
nop
nop
rcall
        DASH
        DOT
rcall
rcall
        DOT
rcall
        DOT
ret
rcall
        DASH
rcall
        DOT
rcall
        DASH
rcall
        DOT
ret
rcall
        DASH
rcall
        DOT
rcall
        DOT
ret
nop
rcall
        DOT
ret
nop
nop
nop
rcall
        DOT
rcall
        DOT
rcall
        DASH
rcall
        DOT
ret
rcall
        DASH
rcall
        DASH
rcall
        DOT
ret
nop
rcall
        DOT
rcall
        DOT
rcall
        DOT
rcall
        DOT
ret
rcall
        DOT
        DOT
rcall
ret
nop
nop
```

rcall

DOT

```
rcall
        DASH
        DASH
rcall
rcall
        DASH
ret
rcall
        DASH
        DOT
rcall
rcall
        DASH
ret
nop
        DOT
rcall
rcall
        DASH
rcall
        DOT
rcall
        DOT
ret
rcall
        DASH
rcall
        DASH
ret
nop
nop
        DASH
rcall
rcall
        DOT
ret
nop
nop
rcall
        DASH
rcall
        DASH
        DASH
rcall
ret
nop
rcall
        DOT
rcall
        DASH
rcall
        DASH
rcall
        DOT
ret
rcall
        DASH
rcall
        DASH
        DOT
rcall
rcall
        DASH
ret
rcall
        DOT
rcall
        DASH
rcall
        DOT
ret
nop
```

rcall

DOT

```
rcall
        DOT
 rcall
        DOT
 ret
 nop
 rcall
        DASH
 ret
 nop
 nop
 nop
 rcall
        DOT
 rcall
        DOT
 rcall
        DASH
 ret
 nop
 rcall
        DOT
 rcall
        DOT
 rcall
        DOT
 rcall
        DASH
 ret
 rcall
        DOT
        DASH
 rcall
 rcall
        DASH
 ret
 nop
 rcall
        DASH
 rcall
        DOT
 rcall
        DOT
 rcall
        DASH
 ret
 rcall
        DASH
 rcall
        DOT
 rcall
        DASH
 rcall
        DASH
 ret
 rcall
        DASH
 rcall
        DASH
 rcall
        DOT
 rcall
        DOT
 ret
; *********************
;* Additional Program Includes
; *********************
.include "LCDDriver.asm" ; LCD stuff
```

```
;* Stored Program Data
WELCOME L1 S:
   "Welcome!
.DB
WELCOME_L1_E:
WELCOME L2 S:
   "Please press PDO"
.DB
WELCOME_L2_E:
PROMPT_S:
   "Enter word:
.DB
PROMPT E:
; *********************
;* Data Memory Allocation
.dseg
.org $0100
LCD_Line1: .byte $10
.org $0110
LCD_Line2: .byte $10
.org $0120
LCD_End:
```